q-類似のCoqによる形式化

アドバイザー: Jacques Garrigue 教授

学籍番号: 322101289 氏名:中村 薫

ът I 11 m

2023年1月6日

序文

本論文の主結果は、q-類似の初等的な結果を Coq によって形式化するものである。具体的には Victor Kac, Pokman Cheung の Quantum Calculus [1] の 4 章 (4.1) 式の q-Taylor 展開,及びその系と して得られる Gauss's binomial formula の形式化を目標としている。本論文での q-類似に関する定義や定理、証明は [1] によるものだが、その形式化を行ったという点において独自性がある。形式化 したコード全体は https://github.com/nakamurakaoru/q-analogue [2] にある.

q-類似は、学部 4 年次に卒業研究のテーマとして扱ったものである。 実数パラメータ q、実数上の関数 f に対して

$$D_q f(x) := \frac{f(qx) - f(x)}{(q-1)x}$$

で定義される q-微分を出発点とし、この q-微分に対してうまく振る舞い、かつ q を極限で 1 に近づけると通常の定義に一致するように数学の諸概念を一般化するものである。例えば、自然数 n について x^n を定義に沿って q-微分すると、

$$D_q x^n = \frac{(qx)^n - x^n}{(q-1)x} = \frac{q^n - 1}{q-1} x^{n-1}$$

となる. 通常の微分では, $(x^n)' = nx^{n-1}$ となることと比較して, $n \circ q$ -類似 [n] を

$$[n] = \frac{q^n - 1}{q - 1}$$

と定める. また, $(x-a)^n$ の q-類似は,

$$(x-a)_q^n := \begin{cases} 1 & (n=0) \\ (x-a)(x-qa)\cdots(x-q^{n-1}a) & (n \ge 1) \end{cases}$$

と定義することで、q-微分と自然数のq-類似に対してうまく振る舞う、つまり

$$D_q(x-a)_q^n = [n](x-a)_q^{n-1}$$

が成り立つ. 更に、階乗と二項係数のq-類似をそれぞれ

$$[n]! := \begin{cases} 1 & (n=0) \\ [n] \times [n-1] \cdots [1] & (n \ge 1) \end{cases}$$
$$\begin{bmatrix} n \\ j \end{bmatrix} := \frac{[n]!}{[j]![n-j]!}$$

で定めれば、q-Taylor 展開、Gauss's binomial formula は以下のように書ける.

- *q*-Taylor 展開 -

f(x) を N 次の実係数多項式とする. 任意の $c \in \mathbb{R}$ について

$$f(x) = \sum_{j=0}^{N} (Dq^{j}f)(c) \frac{(x-c)_{q}^{j}}{[j]!}$$

が成り立つ.

Gauss's binomial formula

 $x, a \in \mathbb{R}, n \in \mathbb{Z}_{>0}$ について

$$(x+a)_q^n = \sum_{i=0}^n \begin{bmatrix} n \\ j \end{bmatrix} q^{\frac{j(j-1)}{2}} a^j x^{n-j}$$

が成り立つ.

この2つを形式化することが本論文の主目的である.

Coq とは、定理証明支援系の1つであり、数学的な証明が正しいかどうか判定するプログラムである。人間がチェックすることが難しい複雑な証明でも正しさが保証され、また証明付きプログラミングにも応用される。 Mizar, Isabelle/HOL 等他にも定理証明支援系は存在するが、修士1年次後期に履修した授業で Coq の使い方を学んだため、今回の形式化に利用した。 実際に Coq が用いられた有名な例として、四色定理や Feit-Thompsonの定理 (奇数位数定理) などがある。 Coq は型付き λ 計算という理論に基づいている。 今回の証明に関しては、 Coq の標準ライブラリ [4] に加えて、数学の証明のために整備されたライブラリ群である mathcomp [5] も用いている。

今後の展望としては、まずはこれまでに形式化した q-類似の各概念が $q \to 1$ としたときに通常の数学の概念に一致することの形式化を行いたい。このためには、現在開発中のライブラリである mathcomp analysis [6] を用いる必要がある。また、このライブラリを用いると無限和に関する形式化も可能であるため、Gauss's binomial formula を無限に拡張したものや、無限和を用いて定義される指数関数、三角関数の q-類似の形式化にも挑戦していきたい。

最後に構成について述べる。まず 1 節で q-類似の概要について説明し、2 節で修士 1 年次に少人数クラスで学習した H.P.Barendregt の Lambda Calculi with Types [8] に沿って、型付き λ 計算について述べる。3 節は Coq の概要についてで、特に 3.2 節で Coq や mathcomp の使い方を具体例を交えて説明するが、より詳細な情報については萩原 学/アフェルト・レナルドの Coq/SSReflect/Mathcomp [3] 等を参照のこと。これらの準備のもと、4 節から本題の形式化に入る。[1] での定義、定理を述べた後、その形式化を与え、必要であれば形式化をするにあたっての注意点を述べることを繰り返すという流れである。証明の方針等は基本的に [1] の通りであるが、4.3 節では一部 [1] から離れ、多項式として q-微分や q-二項式を定義しなおして形式化を行っている。これらの新たな定義が多項式に対してのもとの定義を適用したものと一致していることの証明も行っている。

目次

1	q-類似	3
2	型付きん計算	4
	2.1 λ計算	4
	$2.2 \lambda \rightarrow \lambda \lambda 2 \dots \dots \dots$	4

3	Coq	4
	3.1 λ-cube と CIC	4
	3.2 Coq の使い方	4
	3.3 mathcomp の型	9
4	形式化	9
	4.1 <i>q</i> -微分の定義	0
	4.2 $(x-a)^n$ の q -類似	1
	4.3 関数から多項式へ 1	8
	4.4 <i>q</i> -Taylor 展開	21

1 q-類似

q-類似とは, $q \to 1$ とすると通常の数学に一致するような拡張のことである. 例えば, 自然数 n の q-類似 [n] は

$$[n] = 1 + q + q^2 + \cdots + q^{n-1}$$

であり、 $(x-a)^n$ の q-類似 $(x-a)_q^n$ は

$$(x-a)_q^n := \begin{cases} 1 & (n=0) \\ (x-a)(x-qa)\cdots(x-q^{n-1}a) & (n \ge 1) \end{cases}$$

である. 本論文では、この $(x-a)_q^n$ に対して、

$$(x+a)_q^n = \sum_{j=0}^n \begin{bmatrix} n \\ j \end{bmatrix} q^{\frac{j(j-1)}{2}} a^j x^{n-j}$$

が成り立つことの形式化を目標としている。あえてパラメータを増やすq-類似を考える利点の一つとしては、証明が複雑な定理に対してより簡単な別証明を与えられる場合があることである。例えば、Jacobi の三重積 ([1] p35 Theorem 11.1)

 $z, q \in \mathbb{R}, |q| < 1 \ge \cup \mathcal{T},$

$$\sum_{n=-\infty}^{\infty} q^{n^2} z^n = \prod_{n=1}^{\infty} (1 - q^{2n})(1 + q^{2n-1}z)(1 + q^{2n-1}z^{-1})$$

が成り立つ.

はその一例である. 楕円関数論の文脈で登場する恒等式であるが ([7] p144 (3.47) 等を参照), q-類似で得られる式

$$(1+x)_q^{\infty} = \sum_{j=0}^{\infty} q^{j(j-1)/2} \frac{x^j}{(1-q)(1-q^2)\cdots(1-q^j)} \quad ([1] \text{ p30 } (9.3) \ \vec{\Xi})$$

$$\frac{1}{(1-x)_q^{\infty}} = \sum_{i=0}^{\infty} \frac{x^j}{(1-q)(1-q^2)\cdots(1-q^j)} \quad ([1] \text{ p30 } (9.4) \ \vec{\Xi})$$

を用いることで簡単に証明できる.

2 型付きλ計算

2.1 λ計算

まず型のない λ 計算を定義する. 初めに, λ 計算がどのようなものなのかについての概要を説明 し、その後厳密な定義に移る.

λ計算の概要

 λ 計算には、抽象と適用の2つの基本的な操作がある。まず、抽象については、「式から関数を作る操作」と捉えることができる。M を λ 計算における式 (λ 計算においてはこれを λ 項と呼ぶ) だとすると、

$\lambda x.M$

- $\lambda x.(x^2 + 3xy + 4)$
- $\lambda xy.(x^2 + 3xy + 4)$
- $\lambda z.(x^2 + 3xy + 4)$

はそれぞれ、1つ目は $x \mapsto (x^2 + 3xy + 4)$ というxについての2次関数、2つ目は $(x,y) \mapsto (x^2 + 3xy + 4)$ というx,yについての2変数関数を表す。3つ目は、 $(x^2 + 3xy + 4)$ は変数zを含まないので、定数関数を表すことになる。

もう1つの操作である適用は、2つの λ 項MとNを並べて、

MN

と書かれ, 直観的には「関数 M に値 N を代入する」ことを示している. 例えば, M が $\lambda x.(3x+2), N$ が 4 であれば、

$$(\lambda x.(3x + 2)) 4 = 3 \cdot 4 + 2 (= 14)$$

となる. 一般には, [x := N] で x に N を代入することを表すとして,

$$(\lambda x.M) N = M[x := N]$$

と書く.

λ計算の定義

- 2.2 $\lambda \rightarrow \xi \lambda 2$
- 3 Coq
- 3.1 λ -cube \succeq CIC
- 3.2 Coq の使い方

この節では Coq のコマンドとタクティックの使い方について述べる. ここでは, タクティックは 証明の中でコンテクスト (変数や仮定) やゴール (証明すべき主張) を変形させるもの, コマンドは タクティック以外のものとして扱う. まず, よく使うコマンドについて説明する.

Require Import ライブラリを読み込むためのコマンドである. From mathcomp Require Import ssreflect であれば, ライブラリ群 mathcomp から ssreflect を読み込んでいる.

Section / End Section [セクション名], End [セクション名] でセクションを作ることができ、そのセクション内共通のコンテキストを宣言できる.

Variable Variable [変数]: [型]で,特定の型を持つ変数を宣言できる. 例えば

Variable n: nat

で、変数 n が自然数型 nat の要素であることを表している. Section/End コマンドと組み合わせることで、End まで同じ意味で扱われ、End 以降は効力を失う. 同時に複数の変数を宣言することもできる. その場合は

Variables [変数] [変数] … [変数]: [型]

と書く(ただし, Coq にとっては Variable と Variables に違いは無い).

Hypothesis Hypothesis [仮定名]: [仮定]で仮定を置くことができる. Variable 同様, Section/End と組み合わせることで, セクション内共通の仮定を置くことができる.

Definition 新たに関数を定義するためのコマンドで、

Definition [関数名] ([引数]: [引数の型]): [関数の型] := [関数の定義式] という形で用いる.

Definition dq (f : $R \rightarrow R$) x := f (q * x) - f x.

であれば、dq が定義の名前、f、x が引数、 $R \rightarrow R$ が f の型であり、f (q * x) - f x が dq を定義する式である。 また、x と dq そのものの型は推論できるため省略できる。

Fixpoint 再起関数を定義するためのコマンドで、

Fixpoint [関数名] ([引数]: [引数の型]): [関数の型] := [定義中の関数を含む定義式] と書く. 停止しない関数を認めてしまうと矛盾が生じるため, 停止性が保証されていない関数を定義することはできない.

Lemma 補題を宣言するためのコマンドで、

Lemma [補題名] ([引数]: [引数の型]): [補題の主張] という形である.

Lemma Dq_pow n x : $x \neq \emptyset \rightarrow Dq$ (fun $x \Rightarrow x \hat{n}$) $x = qnat n * x \hat{n}$ (n - 1).

であれば、Dq_pow が補題名、n,x が引数、:以降が補題の主張である.

Lemma の代わりに Theorem, Corollary 等でも同じ機能をもつ.

Proof/Qed Proof は Lemma の後に書いて補題の主張と証明を分ける (実際には省略可能で, 人間の見やすさのために書いている). 証明を完了させて Qed を書くことで Coq に補題を登録することができ, 他の補題の証明に使えるようになる.

次に、タクティックについて述べる. よく使われるタクティックは move, apply, rewrite の 3 つである.

move move=> H でゴールの前提に H という名前をつけてコンテクストに移動する. また move: H で補題 H もしくはコンテクストに存在する H をゴールの前提に移す.

apply 補題 lem が P1 \rightarrow P2 という形で, ゴールが P2 のとき, apply lem でゴールを P1 に変える. コンテクストの仮定 H が P1 \rightarrow P2 であれば apply H で同じことができる.

rewrite defが定義のとき, rewrite /defでゴールに出現している defを展開する.

また、補題 lem が A = B という形のとき、rewrite lem でゴールに出現する A を B に書き換える (ただし、lem が $H \rightarrow (A = B)$ という形であるとき、H がゴールに追加される). 更に、rewrite lem in H で、コンテクストのH に出現する A を B に書き換える. apply と同じく、使いたい等式が仮定にある場合も同じように使える.

以下、2つの具体例を用いて Proof 内でのタクティックの使い方を説明する.

Example 3.2.1 モーダスポーネンス

命題 P, Q について, $P \Longrightarrow Q$ かつ P であれば, Q が成り立つということを Coq で証明する. まずこの主張を形式化すると以下の通り.

From mathcomp Require Import ssreflect.

Lemma modus_ponens (P Q : Prop) : $(P \rightarrow Q) \land P \rightarrow Q$.

Prop とは **Coq** において命題全体を表す型であり, // は「かつ」を表している このとき, **Coq** のゴールエリア (コンテクストとゴールが表示される画面) は以下の通りである.

```
1 subgoal
P, Q : Prop
(P \rightarrow Q) \land P \rightarrow Q
```

---の上がコンテクスト,下がゴールである. 1 行目の 1 subgoal はゴールが 1 つであることを示しており, 用いるタクティックによってはゴールが増えることもある. 命題 P_1 , P_2 , P_3 について $P_1/\backslash P_2 \to P_3$ と $P_1 \to P_2 \to P_3$ は同じ意味であり,この書き換えは move=> [] で行える. 実行する と以下のようにゴールエリアが変化する.

```
1 subgoal
P, Q: Prop
(P \rightarrow Q) \rightarrow P \rightarrow Q
```

ゴールに前提 $(P \rightarrow Q)$ があるため, move=> pq で pq という名前をつけてコンテクストに移動する.

```
1 subgoal
P, Q: Prop
pq: P \rightarrow Q

P \rightarrow Q
```

まだ前提 P があるため, move=> p で p と名付けてコンテクストに移動する.

```
1 subgoal
P, Q : Prop
pq : P → Q
p : P

Q
```

ゴールが Q であり, コンテクストに P \rightarrow Q という仮定 pq があるので, apply pq でゴールを P に 書き換える.

```
1 subgoal
P, Q : Prop
pq : P → Q
p : P
------
P
```

ここまで来ると, ゴールが P であり, コンテクストに P があるため, by [] で証明を終了する. done も by [] と同じ意味を持つ.

No more subgoals.

ゴールエリアに No more subgoals. と表示されれば証明は終了であり, Qed を書くことで補題として登録されることになる.

以上をまとめると次のようになる.

```
Lemma modus_ponens (P Q : Prop) : (P \rightarrow Q) \land P \rightarrow Q. Proof.  
move \Rightarrow [].  
move \Rightarrow pq.  
move \Rightarrow p.  
apply pq.  
by []. Qed.
```

説明のため細かく 1 行づつ書いたが、複数の move はまとめられること、あるタクティックにより ゴールが自明なもの (コンテクストに存在する、A=A である、計算から簡単に示されるなど) になる場合はそのタクティックの前に by をつけることで証明を終了させられることを用いれば、次のように短くすることができる.

```
Lemma modus_ponens (P Q : Prop) : (P \rightarrow Q) \land P \rightarrow Q. Proof. 
 move \Rightarrow [] pq p. 
 by apply pq. 
 Qed.
```

Coq による証明は、Curry-Howard 同型と呼ばれる、

命題 ↔ 型 証明 ↔ 型に要素が存在する

という対応関係に基づいている.また,論理演算子についても,以下のような対応がある.

$$P$$
ならば Q $P \rightarrow Q$ P

この同型をもとに上記の証明をもう一度考えてみると, $P \to Q$ と Pという型に要素が存在することから, Qという型の要素を構成すればよいということである.

まず, 前提の要素それぞれに pq, p と名前をつける. これがプログラム中の $move \Rightarrow []$ pq p のことである. ここで, $P \to Q$ という型は, 入力する値の型が P, 出力する値の型が Q であるような関数

の型であるため, P の要素 p に pq を適用することで, Q の要素を構成することができる. この関数 適用がプログラム中の apply pq のことである.

Example 3.2.2 代入計算

自然数m,nについて,

$$m = 0 \Longrightarrow n + m = n$$

という簡単な代入に関する計算を証明してみる.まず主張を形式化する.

Lemma substitution m n : $m = 0 \rightarrow n + m = n$.

このとき, ゴールエリアは以下の通りである.

```
1 subgoal m, n : nat -----
m = 0 \rightarrow n + m = n
```

まず m = 0という前提を move=> Hm で Hm という名前をつけてコンテクストに移動する.

```
1 subgoal
m, n : nat
Hm : m = 0
-----
n + m = 0
```

仮定にm = 0という等式があるため、rewrite Hm でゴールのmを0に書き換える.

```
1 subgoal
m, n : nat
Hm : m = 0
-----
n + 0 = n
```

次に, n + 0 を n に書き換えたい. mathcomp の ssrnat に addn0 という,

forall n : nat, n + 0 = n

に対応する補題があるため、rewrite addnoを実行する.

```
1 subgoal
m, n : nat
Hm : m = 0
-----
n = n
```

このとき, ゴールは同じもの同士の等号であるため, 自明に成り立つ, つまり by [](もしくは done) で終了する.

No more subgoals.

この証明をまとめると以下の通りである.

```
Lemma substitution m n : m = 0 \rightarrow n + m = n. 
 Proof. 
 move\Rightarrow Hm. 
 rewrite Hm. 
 rewrite addn0. 
 by []. 
 Qed.
```

複数の rewrite がまとめられることと by の使い方から, より短く次のように書ける.

Lemma substitution m n : $m = 0 \rightarrow n + m = n$. Proof.

move⇒ Hm.

by rewrite Hm addn0.

Oed

更に, 前提が A = B の形であるとき, $move \rightarrow$ でゴールの $A \in B$ に書き換えられること, ; で異なる タクティックをつなげられることを用いれば以下のように書くこともでできる.

```
Lemma substitution m n : m = 0 \rightarrow n + m = n.
Proof. by move\rightarrow; by rewrite addn0. Qed.
```

Remark 3.2.3 正確には, addn0 は

right_id 0 addn

という補題であり、addn は自然数同士の加法である. right_id の定義は

```
fun (S T : Type) (e : T) (op : S \rightarrow T \rightarrow S) \Rightarrow \forall x : S, op x e = x
```

であり、単位元を右から作用させても元のままであるということを一般的に定義している. addn0 は right_id の e に 0 を、opに addn を入れたものであるので、

```
\forall x : nat, addn x \emptyset = x となる.
```

このように、Coq での証明は、ゴールを自明な形になるまで繰り返し書き変えていくやり方が基本である.

3.3 mathcomp の型

4 形式化

この定義を形式化するためまずは実数の形式化を考える。今回は mathcomp の ssrnum にある rcfType を用いる. rcfType とは, real closed field, つまり実閉体全体からなる型であり, この型の 要素を実数として用いることにする. mathcomp はある型を構成するために他の型を用いているため, ヒエラルキー (階層構造) がある。通常の数学において体が環の性質を, 環が群の性質を引き継 ぐように, mathcomp でもより一般の型の性質を引き継いでいる. rcfType も多くの性質を引き継いでおり, 特に環の性質, ringType を持っていることが重要である。実際, 形式化で利用するライブラリの補題の多くは ringType に対するものである.

自然数型 nat .+1 は nat での後続数を表す 整数型 int

4.1 *q*-微分の定義

様々な q-類似を考えるにあたって, まずは微分の q-類似から始める. 以下, q を 1 でない実数とする.

Definition 4.1.1 ([1] p1 (1.1), p2 (1.5)) 関数 $f: \mathbb{R} \to \mathbb{R}$ に対して, f(x) の q 差分 $d_q f(x)$ を,

$$d_q f(x) \coloneqq f(qx) - f(x)$$

と定める. 更に, f(x) の q 微分を $D_a f(x)$ を,

$$D_q f(x) := \frac{d_q f(x)}{d_q x} = \frac{f(qx) - f(x)}{(q-1)x}$$

と定める.

rcfType を使って q-微分の形式化を以下のように行う.

From mathcomp Require Import all_ssreflect all_algebra. Import GRing.

Section q_analogue.

Local Open Scope ring_scope.

Variables (R : rcfType) (q : R).

Hypothesis Hq : $q - 1 \neq 0$.

Notation "f // g" := (fun x \Rightarrow f x / g x) (at level 40).

Definition dq (f : R \rightarrow R) x := f (q * x) - f x.

Definition Dq f := dq f // dq id.

5 行目の Variables (R: rcfType) (q: R). は, R が rcfType の要素であること, q が R の要素であることを表している. rcfType は実閉体全体を表す型であるため, R は具体的な型ではなく, 実閉体としての性質をもつ抽象的な型として扱っている.

Remark 4.1.2 *f* が微分可能であるとき,

$$\lim_{q \to 1} D_q f(x) = \frac{d}{dx} f(x)$$

が成り立つが,本論文においては極限操作に関しての形式化は扱わない.

次に, x^n $(n \in \mathbb{Z}_{\geq 0})$ を q-微分した際にうまく振る舞うように自然数の q-類似を定義する.

Definition 4.1.3 ([1] p2 (1.9)) $n \in \mathbb{Z}_{>0}$ に対して, n の q-類似 [n] を,

$$[n] := \frac{q^n - 1}{q - 1}$$

と定義する.

この [n] に対して, $(x^n)' = nx^{n-1}$ の q-類似が成り立つ.

Proposition 4.1.4 ([1] p2 Example (1.7)) $n \in \mathbb{Z}_{>0}$ について,

$$D_a x^n = [n] x^{n-1}$$

が成り立つ.

Proof. 定義に従って計算すればよく、

$$D_q x^n = \frac{(qx)^n - x^n}{(q-1)x} = \frac{q^n - 1}{q-1} x^{n-1} = [n] x^{n-1}$$

П

この定義と補題の形式化は以下のとおりである.

Definition qnat $n : R := (q \cdot n - 1) / (q - 1)$.

Lemma Dq_pow n x : $x \neq \emptyset \rightarrow Dq$ (fun $x \Rightarrow x \hat{n}$) $x = qnat n * x \hat{n} (n - 1)$. Proof.

 $move \Rightarrow Hx$.

rewrite /Dq /dq /qnat.

rewrite -{4}(mul1r x) -mulrBl expfzMl -add_div; last by apply mulf_neq0.

rewrite $[in x ^n](_ : n = (n -1) +1) //;$ last by rewrite subrK.

rewrite expfzDr ?expr1z ?mulrA -?mulNr ?red_frac_r ?add_div //.

rewrite -{2}[x ^ (n - 1)]mul1r -mulrBl mulrC mulrA.

by rewrite [in $(q - 1)^-1 * (q ^ n - 1)$] mulrC.

証明で用いている補題について, 例えば mul1r は

∀ R : ringType, right_id 1 *%R

という補題であり、任意の ringType を型にもつ R について、R 上の単位元の性質を表している。ここで、補題の R の型は rcfType ではなく ringType であるが、前述の通り rcfType は ringType の性質を持っているため、今回の形式化に用いている rcfType 型の R についてもこの補題を使うことができる。補題名 mullrの r は ringType に対する補題であることを表しており、mulrBl、subrK、mulrC なども同じである。

また red_frac_r は,

 \forall x y z : R, z \neq 0 \rightarrow x * z / (y * z) = x / y

という自分で用意した補題である (このように直接 q-類似に関係はしないが, 形式化をするために用意した補題は [2] に q-tool.v というファイルにまとめている). この補題の本質は z/z=1 という約分計算であり, mathcomp の ssralg の補題

divff : \forall (F : fieldType) (x : F), $x \neq 0 \rightarrow x / x = 1$

を用いているため $\mathbf{z} \neq \mathbf{0}$ という仮定が必要になる. よって, \mathbf{Dq} -pow にも $\mathbf{z} \neq \mathbf{0}$ という前提を加えている. 今後も補題を形式化するにあたって, その証明の中で約分を行う際には $\mathbf{0}$ でないという前提を付け加えることになる.

Remark 4.1.5 qnat という名前であるが、実際には n の型は nat ではなく int にしている. また、 Dq_-of_-pow の n の型は int であるため、より一般化した形での形式化になっている.

[1] では証明は 1 行で終わっているが、形式化する場合には何倍もかかっている。これは、積の交換法則や指数法則などの、通常の数学では当たり前なことが自動では計算されず、rewrite mulrCや rewrite expfzDrというように rewrite expfzDrというように rewrite expfzDrというように expfzDrというと expfzDrというように expfzDrというように expfzDrというと expfzDrというと expfzDrというと expfzDrというと expfzDrというと expfzDrというと expfzDrというと expfzDrというと expfzDrとの expfzDrとの

4.2 $(x-a)^n$ の q-類似

続いて $(x-a)^n$ の q-類似を定義し, その性質を調べる.

Definition 4.2.1 ([1] p8 Definition (3.4)) $x, a \in \mathbb{R}, n \in \mathbb{Z}_{\geq 0}$ に対して, $(x - a)^n$ の q-類似 $(x - a)^n_q$ を,

$$(x-a)_q^n = \begin{cases} 1 & \text{if } n = 0\\ (x-a)(x-qa)\cdots(x-q^{n-1}a) & \text{if } n \ge 1 \end{cases}$$

と定義する.

Proposition 4.2.2 $n \in \mathbb{Z}_{>0}$ に対し,

$$D_q(x-a)_q^n = [n](x-a)_q^{n-1}$$

П

が成り立つ.

Proof. n についての帰納法により示される.

まず、 $(x-a)_a^n$ の定義を形式化する.

Fixpoint qbinom_pos a n x := match n with

 $\mid 0 \Rightarrow 1$ $\mid n0.+1 \Rightarrow (qbinom_pos a n0 x) * (x - q ^ n0 * a)$ end.

Fixpoint を用いて再帰的な定義をしており、match を使ってnが0かどうかで場合分けしている。再帰で呼び出す際に引数nが真に小さくなっているため停止性が保証されている。補題の証明については以下の通り。

```
Theorem Dq_qbinom_pos a n x : x \neq 0 \rightarrow
  Dq (qbinom_pos a n.+1) x =
  qnat n.+1 * qbinom_pos a n x.
Proof.
  move \Rightarrow Hx.
  elim: n \Rightarrow [|n|]H.
  - rewrite /Dq /dq /qbinom_pos /qnat.
    rewrite !mul1r mulr1 expr1z.
    rewrite opprB subrKA !divff //.
    by rewrite denom_is_nonzero.
  - rewrite (_ : Dq (qbinom_pos a n.+2) x =
                  Dq ((qbinom_pos a n.+1) **
                  (\text{fun } x \Rightarrow (x - q \hat{ (n.+1) * a)}) x) //.
    rewrite Dq_prod' //.
    rewrite [Dq (+%R^{^{^{\prime}}} (-(q^n.+1*a))) x]/Dq/dq.
    rewrite opprB subrKA divff //; last by apply denom_is_nonzero.
    rewrite mulr1 exprSz.
    rewrite -[q * q ^ n * a]mulrA -(mulrBr q) IH.
    rewrite -[q * (x - q ^ n * a) * (qnat n.+1 * qbinom_pos a n x)]mulrA.
    rewrite [(x - q ^ n * a) * (qnat n.+1 * qbinom_pos a n x)]mulrC.
    rewrite -[qnat n.+1 * qbinom_pos a n x * (x - q ^ n * a)]mulrA.
    rewrite (_ : qbinom_pos a n x * (x - q \hat{n} * a) = qbinom_pos a n.+1 x) //.
    rewrite mulrA -{1}(mul1r (qbinom_pos a n.+1 x)).
    by rewrite -mulrDl -qnat_cat1.
0ed.
```

ここで elim: n は n の帰納法に対応している.

指数法則については、一般には $(x-a)^{m+n} \neq (x-a)_q^m (x-a)_q^n$ であり、以下のようになる.

Proposition 4.2.3 ([1] p8 (3.6)) $x, a \in \mathbb{R}, m, n \in \mathbb{Z}_{>0}$ について,

$$(x-a)_q^{m+n} = (x-a)_q^m (x-q^m a)_q^n$$

が成り立つ.

Proof.

$$(x-a)_q^{m+n} = (x-a)(x-qa)\cdots(x-q^{m-1}a)\times(x-q^ma)(x-q^{m+1}a)\cdots(x-q^{m+n-1})$$

$$= (x-a)(x-qa)\cdots(x-q^{m-1}a)\times(x-q^ma)(x-q(q^mx))\cdots(x-q^{n-1}(q^ma))$$

$$= (x-a)_q^m(x-q^ma)_q^n$$

より成立する.

この形式化は次のとおりである.

Lemma qbinom_pos_explaw x a m n :

 $qbinom_pos a (m + n) x =$

qbinom_pos a m x * qbinom_pos (q ^ m * a) n x.

Proof.

elim: n.

- by rewrite addn0 /= mulr1.
- elim \Rightarrow [_|n _ IH].
 - + by rewrite addnS /= addn0 expr0z !mul1r.
 - + rewrite addnS [LHS]/= IH /= !mulrA.
 by rewrite -[q ^ n.+1 * q ^ m] expfz_n0addr // addnC.

Qed.

[1] の証明では単に式変形しているが、形式化の証明では \mathbf{m} , \mathbf{n} に関する帰納法を用いている. この指数法則を用いて、 $(x-a)_q^n$ の n を負の数に拡張する. まず、[1] の定義は

Definition 4.2.4 ([1] p9 (3.7)) $x, a \in \mathbb{R}, l \in \mathbb{Z}_{>0}$ とする. このとき,

$$(x-a)_q^{-l} := \frac{1}{(x-q^{-l}a)_q^l}$$

と定める.

であり、この形式化は、

Definition qbinom_neg a n x := 1 / qbinom_pos (q $\hat{}$ ((Negz n) + 1) * a) n x.

となる. ここで, Negz n とは Negz n = - n.+1 をみたすものであるので, (Negz n) + 1 は-n となり, [1] の定義と一致する. また int は

Variant int : Set := Posz : nat \rightarrow int | Negz : nat \rightarrow int.

のように定義されている. よって, int は 0 以上か負かで場合分けできるため, n: int に対して qbinom_pos の定義を以下のように整数に拡張する.

Definition qbinom a n x :=

match n with

| Posz n0 \Rightarrow qbinom_pos a n0 x

| Negz n0 \Rightarrow qbinom_neg a n0.+1 x

end.

整数に拡張した $(x-a)_q^n$ についても, 指数法則と q-微分はうまく振る舞う. まず指数法則について見ていく.

Proposition 4.2.5 ([1] p10 Proposition 3.2) $m, n \in \mathbb{Z}$ について, Proposition 4.2.3 は成り立つ.

Proof. m,n の正負で場合分けして示す. m>0 かつ n>0 の場合はすでに示しており, m=n=0 の場合は定義からすぐにわかる. その他の場合について, まず m<0 かつ $n\geq0$ の場合, m=-m' とおくと

$$(x-a)_q^m (x-q^m)_q^n = (x-a)_q^{-m'} (x-q^{-m'}a)_q^n$$

$$= \frac{(x-q^{-m'}a)_q^n}{(x-q^{-m'}a)_q^{m'}}$$

$$= \begin{cases} (x-q^{m'}(q^{-m'}a))_q^{n-m'} & n \ge m' \\ \frac{1}{(x-q^n(q^{-m'}a))_q^{m'-n}} & n < m' \end{cases}$$

$$= (x-a)_q^{n-m'}$$

$$= (x-a)_q^{n+m}$$

というように, n と m' の大小で場合分けすることで示せる. 次に, $m \ge 0$ かつ n < 0 の場合, n = -n' として.

$$(x-a)_q^m (x-q^m)_q^n = (x-a)_q^m (x-q^m a)_q^{-n'}$$

$$= \begin{cases} \frac{(x-a)_q^{m-n'} (x-q^{m-n'} a)_q^{n'}}{(x-q^{m-n'} a)_q^n} & m \ge n' \\ \frac{(x-a)_q^m}{(x-q^{m-n'})_q^{n'-m} (x-q^{n'-m} (q^{m-n'} a))} & m < n \end{cases}$$

$$= \begin{cases} (x-a)^{m-n'} & m \ge n' \\ \frac{1}{(x-q^{m-n'})_q^{n'-m}} & m < n' \end{cases}$$

$$= (x-a)_q^{m-n'} = (x-a)_q^{m+n}$$

となる. 最後に, m < 0 かつ n < 0 のとき, m = -m', n = -n' として,

$$(x-a)_q^m (x-q^m)_q^n = (x-a)_q^{-m'} (x-q^{-m'})_q^{-n'}$$

$$= \frac{1}{(x-q^{-m'}a)_q^{m'} (x-q^{-n'-m'}a)_q^{n'}}$$

$$= \frac{1}{(x-q^{-n'-m'}a)_q^{n'} (x-q^{n'}(q^{-m'-n'}a))_q^{m'}}$$

$$= \frac{1}{(x-q^{-n'-m'}a)_q^{n'+m'}}$$

$$= (x-a)_q^{-m'-n'}$$

$$= (x-a)_q^{m'+n'}$$

となる.

この補題を形式化すると次のようになる.

Theorem qbinom_explaw a m n x : $q \neq 0 \rightarrow$ qbinom_denom a m x $\neq 0 \rightarrow$ qbinom_denom (q ^ m * a) n x $\neq 0 \rightarrow$ qbinom a (m + n) x = qbinom a m x * qbinom (q ^ m * a) n x. Proof.

 $move \Rightarrow Hq0$.

case: $m \Rightarrow m \text{ Hm}$. - case: $n \Rightarrow n \text{ Hn}$.

```
+ by apply qbinom_pos_explaw.
    + rewrite qbinom_exp_pos_neg //.
      by rewrite addrC expfzDr // -mulrA.
  - case: n \Rightarrow n Hn.
    + by rewrite qbinom_exp_neg_pos.
    + by apply qbinom_exp_neg_neg.
Qed.
証明の構造としては、まず case:mでmが0以上か負かの場合分けを行い、更にそれぞれの場合に
ついて case:nでnの場合分けを行っている.ここで,前提の gbinom_denom の定義は
Definition qbinom_denom a n x :=
match n with
  | Posz n0 \Rightarrow 1
  | Negz n0 \Rightarrow qbinom_pos (q ^ Negz n0 * a) n0.+1 x
  end.
であり、2つの前提は補題の右辺に出現する項の分母が0にならないということである. 証明中に
使われている補題のうち、qbinom_exp_pos_neg、qbinom_exp_neg_pos、qbinom_exp_neg_neg はそ
れぞれm \ge 0 かつn < 0, m < 0 かつn \ge 0, m < 0 かつn < 0 のときの証明の形式化であり、例えば
gbinom_exp_pos_neg については以下の通り.
Lemma qbinom_exp_pos_neg a (m n : nat) x : q \neq 0 \rightarrow
  qbinom_pos (q ^ (Posz m + Negz n) * a) n.+1 x \neq 0 \rightarrow
  qbinom a (Posz m + Negz n) x = qbinom a m x * qbinom (q ^ m * a) (Negz n) x.
Proof.
 move⇒ Hq0 Hqbinommn.
  case Hmn : (Posz m + Negz n) \Rightarrow [1|1] /=.
  - rewrite /qbinom_neg mul1r.
    rewrite (\underline{\phantom{a}}: qbinom_pos a m x = qbinom_pos a (1 + n.+1) x).
      rewrite qbinom_pos_explaw.
      have \rightarrow : q ^ (Negz n.+1 + 1) * (q ^ m * a) = q ^ 1 * a.
        by rewrite mulrA -expfzDr // -addn1 Negz_addK addrC Hmn.
      rewrite -{2}(mul1r (qbinom_pos (q ^ 1 * a) n.+1 x)) red_frac_r.
        by rewrite divr1.
      by rewrite -Hmn.
    apply Negz_transp in Hmn.
    apply (eq_int_to_nat R) in Hmn.
    by rewrite Hmn.
  - rewrite /qbinom_neg.
    have Hmn' : n.+1 = (1.+1 + m)%N.
      move /Negz_transp /esym in Hmn.
      rewrite addrC in Hmn.
      move /Negz_transp /(eq_int_to_nat R) in Hmn.
      by rewrite addnC in Hmn.
    rewrite (_ : qbinom_pos (q ^ (Negz n.+1 + 1) * (q ^ m * a)) n.+1 x = qbinom_pos (q ^ (Negz n.+1 + 1) * (q ^ m * a))
                               (1.+1 + m) x).
      rewrite qbinom_pos_explaw.
      have \rightarrow : q ^ (Negz n.+1 + 1) * (q ^ m * a) =
                q \hat{ } (Negz 1.+1 + 1) * a.
        by rewrite mulrA -expfzDr // !NegzS addrC Hmn.
      have \rightarrow : q ^ l.+1 * (q ^ (Negz l.+1 + 1) * a) = a.
        by rewrite mulrA -expfzDr // NegzS NegzK expr0z mul1r.
      rewrite mulrA.
      rewrite [qbinom_pos (q \hat{} (Negz l.+1 + 1) * a) l.+1 x *
               qbinom_pos a m x]mulrC.
      rewrite red_frac_l //.
      have \rightarrow : a = q ^ 1.+1 * (q ^ (Posz m + Negz n) * a) \Rightarrow //.
```

by rewrite mulrA -expfzDr // Hmn NegzK expr0z mul1r.
apply qbinom_exp_non0r.
 rewrite -Hmn' //.
by rewrite Hmn'.
Qed.

この証明についての注目点としては,

• [1] では *m* と *n'* の大小で場合分けをしていたが, 形式化では,

case Hmn : (Posz m + Negz n) \Rightarrow [1|1] /=.

として, m - n'の値を1とおき、1が0以上かどうかで場合分けをしている

• Coq では A = B という等式はどの型の上でのものなのかが区別されている. eq_int_to_nat という補題は int 上の等式を nat 上の等式に写している.

などが挙げられる.

次に q-微分について見ていく.

Proposition 4.2.6 ([1] p10 Proposition 3.3) $n \in \mathbb{Z}$ $k \supset k \subset N$

$$D_q x^n = [n] x^{n-1}$$

が成り立つ. ただし, n が整数の場合にも, 自然数のときと同様, [n] の定義は

$$\frac{q^n-1}{q-1}$$

である.

Proof. n > 0 のときは Proposition 4.2.2 であり, n = 0 のときは [0] = 0 からすぐにわかる. n < 0 のときは, Definition 4.2.4 と, 商の微分公式の q-類似版である

$$D_q\left(\frac{f(x)}{g(x)}\right) = \frac{g(x)D_q f(x) - f(x)D_q g(x)}{g(x)g(qx)} \quad ([1] \text{ p3 (1.13)})$$

及び Proposition 4.2.2 を用いて示される.

[1] と同じ方針で証明する. まず, n = 0 のときは次の通り.

Lemma Dq_qbinomn0 a x :

Dq (qbinom a 0) x = qnat 0 * qbinom a (-1) x.

Proof. by rewrite Dq_const qnat0 mul0r. Qed.

ここで、Dq_const は

Lemma Dq_const x c : Dq (fun x \Rightarrow c) x = 0.

という定数関数のq-微分は0であるという補題である.次に,n < 0のときは以下のようになる.

Theorem Dq_qbinom_neg a n x : q \neq 0 \rightarrow x \neq 0 \rightarrow (x - q ^ (Negz n) * a) \neq 0 \rightarrow qbinom_pos (q ^ (Negz n + 1) * a) n x \neq 0 \rightarrow Dq (qbinom_neg a n) x = qnat (Negz n + 1) * qbinom_neg a (n.+1) x. Proof.

move⇒ Hq0 Hx Hqn Hqbinom.

destruct n.

- by rewrite /Dq /dq /qbinom_neg /= addrK' qnat0 !mul0r.
- rewrite Dq_quot //.

```
rewrite Dq_const mulr0 mul1r sub0r.
  rewrite Dq_qbinom_pos // qbinom_qx // -mulNr.
  rewrite [qbinom_pos (q \hat{} (Negz n.+1 + 1) * a) n.+1 x *
            (q \hat{n} n.+1 * qbinom_pos (q \hat{n} (Negz n.+1 + 1 - 1) *
               a) n.+1 x)] mulrC.
  rewrite -mulf_div.
  have \rightarrow : qbinom_pos (q ^ (Negz n.+1 + 1) * a) n x /
                 qbinom_pos (q \hat{} (Negz n.+1 + 1) * a) n.+1 x =
                   1 / (x - q ^ (- 1) * a).
    rewrite -(mulr1 (qbinom_pos (q \hat{} (Negz n.+1 + 1) * a) n x)) /=.
    rewrite red_frac_l.
      rewrite NegzE mulrA -expfzDr // addrA -addn2.
      rewrite (\_: Posz (n + 2)%N = Posz n + 2) //.
      by rewrite -{1}(add0r (Posz n)) addrKA.
    by rewrite /=; apply mulnon0 in Hqbinom.
  rewrite mulf_div.
  rewrite -[q ^n.+1 *
             qbinom_pos (q \hat{ } (Negz n.+1 + 1 - 1) * a) n.+1 x *
                (x - q (-1) * a)]mulrA.
  have \rightarrow : qbinom_pos (q ^ (Negz n.+1 + 1 - 1) * a) n.+1 x *
            (x - q (-1) * a) =
            qbinom_pos (q ^ (Negz (n.+1)) * a) n.+2 x \Rightarrow /=.
    have \rightarrow: Negz n.+1 + 1 - 1 = Negz n.+1.
      by rewrite addrK.
    have \rightarrow : q ^ n.+1 * (q ^ Negz n.+1 * a) = q ^ (-1) * a \Rightarrow //.
    rewrite mulrA -expfzDr // NegzE.
    have \rightarrow : Posz n.+1 - Posz n.+2 = -1 \Rightarrow //.
    rewrite -addn1 - [(n + 1).+1]addn1.
    rewrite (\_: Posz (n + 1)%N = Posz n + 1) //.
    rewrite (_ : Posz (n + 1 + 1)%N = Posz n + 1 + 1) //.
    rewrite -(add0r (Posz n + 1)).
    by rewrite addrKA.
  rewrite /qbinom_neg /=.
  rewrite (\_: Negz n.+2 + 1 = Negz n.+1) // -mulf_div.
  congr (_ * _).
  rewrite NegzE mulrC /qnat -mulNr mulrA.
  congr (_ / _).
  rewrite opprB mulrBr mulr1 mulrC divff; last by rewrite expnon0.
  rewrite invr_expz (\_: - Posz n.+2 + 1 = - Posz n.+1) //.
  rewrite -addn1 (_ : Posz (n.+1 + 1)%N = Posz n.+1 + 1) //.
  by rewrite addrC [Posz n.+1 + 1]addrC -{1}(add0r 1) addrKA sub0r.
rewrite qbinom_qx // mulf_neq0 //.
  by rewrite expnon0.
rewrite qbinom_pos_head mulf_neq0 //.
rewrite (_{-}: Negz n.+1 + 1 - 1 = Negz n.+1) //.
  by rewrite addrK.
move: Hqbinom \Rightarrow /=.
move/mulnon0.
by rewrite addrK mulrA -{2}(expr1z q) -expfzDr.
```

非常に長くなっているが積の交換則や結合則などが多く, Dq_quot が商の q-微分公式の形式化であるため, [1] の証明をそのまま形式化したものになっている. また, いくつかの項が 0 でないという条件がついているが, これらの項は Definition 4.2.4 において分母に現れるため, Dq_of_pow のときと同様妥当であると考えられる. これらをまとめて以下のように形式化できる.

```
Theorem Dq_qbinom a n x : q \neq 0 \rightarrow x \neq 0 \rightarrow x - q ^ (n - 1) * a \neq 0 \rightarrow qbinom (q ^ n * a) (- n) x \neq 0 \rightarrow
```

```
Dq (qbinom a n) x = qnat n * qbinom a (n - 1) x.
Proof.
 move \Rightarrow Hq0 Hx Hxqa Hqbinom.
  case: n Hxqa Hqbinom \Rightarrow [|/=] n Hxqa Hqbinom.
  - destruct n.
    + by rewrite Dq_qbinomn0.
    + rewrite Dq_qbinom_pos //.
      rewrite (\_: Posz n.+1 - 1 = n) // -addn1.
      by rewrite (\_: Posz (n + 1)%N = Posz n + 1) ?addrK.
  - rewrite Dq_qbinom_int_to_neg Dq_qbinom_neg //.
        rewrite Negz_addK.
        rewrite ( (n + 1).+1 = (n + 0).+2) //.
        by rewrite addn0 addn1.
      rewrite (\_: Negz (n + 1) = Negz n - 1) //.
      by apply itransposition; rewrite Negz_addK.
    by rewrite Negz_addK addn1.
Qed.
```

case: n で n が 0 以上か負かで場合分けを, **destruct** n で 0 か 1 以上かの場合分けをしており, それぞれの場合で Dq_qbinom_0, Dq_qbinom_pos, Dq_qbinom_neg を使っていることが見て取れる.

4.3 関数から多項式へ

ここまでは [1] の定義に沿って関数に対して q-微分を定義し, また関数として $(x-a)_q^n$ を定義してきたが, q-Taylor 展開や Gauss's binomial formula を形式化するに当たって多項式に対する q-微分と多項式としての $(x-a)_q^n$ を改めて定義する.

ここで、Coqにおける多項式の扱いについて説明する.

- {poly T} で T 係数多項式を表す型となる. T は ringType でなくてはならないが, rcfType は ringType の構造を引き継いでいるため, 今回用いている R に対して {poly R} が定義できる.
- {poly R} は ringType と lmodType の構造を持っている. よって, 前者の性質から今まで使ってきた ringType に対する補題がそのまま使え, 後者からスカラー倍 a *: p(ここで a: R, p: {poly R} である) も定義される.
- \poly_(i < n) E(i) で、次数がn-1次以下,i次の係数がE(i) である多項式表す。
- c%:P で定数 c のみからなる単項式を表す.
- 'Xで変数 x のみからなる単項式を表す.
- p'_i で多項式 pの i 次の係数を表す.
- size pで多項式 pの次数 +1 を表す.
- p.[x] で多項式 pの x での値を表す.

より詳細な内容については mathcomp の poly.v を参照のこと.

この {poly R} を用いて Dq や qbinom を定義し直していく. まず, q-微分について, 多項式に対する d_q を以下のように定義し直す.

```
Definition scale_var (p : {poly R}):= poly_i < size p) (q ^ i * p'_i). Definition dqp p := scale_var p - p.
```

scale_var は多項式 p を受け取り, i 次の係数を q^i 倍した多項式を返す操作である. また, dqp は多項式に対しての d_q と同じ結果になることが確認できる (正確には, dqp を適用した多項式での $\mathbf x$ での値と $\mathbf x \mapsto \mathbf p.[\mathbf x]$ という関数に $\mathbf d_{\mathbf a}$ を適用した関数の $\mathbf x$ での値が等しいということである).

Definition polyderiv (D : (R \rightarrow R) \rightarrow (R \rightarrow R)) (p : {poly R}) := D (fun (x : R) \Rightarrow p.[x]). Notation "D # p" := (polyderiv D p) (at level 49). Lemma dqp_dqE p x : (dqp p).[x] = (dq # p) x.

この dqp を用いて、多項式に対する D_q を定義する.

Definition Dqp p := dqp p %/ dqp 'X.

p%/p' は多項式pを多項式p'で割った商を表している。この定義だけではdqpをdq'Xで割った余りが0でない可能性があるため、q-微分の正しい形式化である保証がない。しかし実際に多項式に対してDqpを計算すると、dqpの定義から、dqp p は定数項が打ち消しあい、またdqp'X は (q-1)*'X となるので割り切れるはずである。よってこのことを証明しておく。

Lemma Dqp_ok p : dqp 'X % | dqp p.

ここで, p'%| pでpがp'で割り切れることを表す. 今後は扱いやすさのため, 'X で約分した形 Definition Dqp'(p: {poly R}) := \poly_(i < size p) (qnat (i.+1) * p'_i.+1).

を用いる. このとき、Dgp と Dgp' が等しいことも示せる.

Lemma Dqp_Dqp'E p : Dqp p = Dqp' p.

また、dqp のときと同様、多項式に対しての D_q と同じであることを確認しておく.

Lemma Dqp'_DqE p x : $x \neq \emptyset \rightarrow (Dqp' p).[x] = (Dq \# p) x.$

Remark 4.3.1 Dqp_Dqp'E には特に条件がなく、Dqp'_DqE には $x \neq 0$ という条件がついている. この違いは、前者は'X / 'X = 1%:P という多項式での約分を、後者は x / x = 1 という実数での約分を行っているということから生じている. 後で詳しく述べるが、約分の際に条件が必要なくなることが多項式で考える利点の一つである.

次に $,(x-a)_a^n$ を多項式として以下のように定義しなおす.

Fixpoint qbinom_pos_poly a n := match n with \mid 0 \Rightarrow 1 \mid n.+1 \Rightarrow (qbinom_pos_poly a n) * ('X - (q ^ n * a)%:P) end.

この多項式のxでの値は元の定義のqbinom_posと等しくなる.

Lemma qbinom_posE a n x :
 qbinom_pos a n x = (qbinom_pos_poly a n).[x].

更に、このように定義した Dqp と qbinom_pos_poly に対しても Proposition 4.2.2 と同じことが成り立つ.

Lemma Dqp'_qbinom_poly a n :
 Dqp' (qbinom_pos_poly a n.+1) = (qnat n.+1) *: (qbinom_pos_poly a n).

Remark 4.3.2 証明の方針はこれまでの関数としての場合と同じだが、 $Dq_prod'(q-微分の積の法則)$ に対応する補題の証明のため、 $scale_var$ が積について分解できること、つまり

Lemma scale_var_prod (p p': {poly R}) : scale_var (p * p') = scale_var p * scale_var p'. を示している. ここで証明の冒頭を抜き出すと以下のようになっている.

```
Proof.
  pose n := size p.
  have : (size p ≤ n)%N by [].
  clearbody n.
  have Hp0 : ∀ (p : {poly R}), size p = 0%N →
      scale_var (p * p') = scale_var p * scale_var p'.
      move⇒ p0 /eqP.
      rewrite size_poly_eq0.
      move/eqP →.
      by rewrite mul0r scale_varC mul0r.
    elim: n p ⇒ [|n IH] p Hsize.
      ...
Qed.
```

pose n := size p. で多項式 p の size を n と置いており, have: (size p <= n)%N by []. で size p が n 以下という自明な主張をあえて置いているが, これは多項式の size に関する帰納法を 用いるためである. このように, 当たり前の内容を明示的に書かなければならないことに加え, 形式 化するための証明の構造を工夫しなければならない場合もある.

多項式で考える理由は, q-Taylor 展開が多項式に対する定理であることに加え, 以下の 2 つの目的がある.

1. x/x = 1 の計算に $x \neq 0$ という条件が必要ない

先に見たように、Coq で約分の計算、つまり x/x=1 を行う際には $x \neq 0$ という条件が必要である。よって、実数 x 対して x/x=1 を計算する場合、後から x=0 を代入することはできない。しかし、多項式で考える場合、'X は単項式であるためゼロ多項式とは異なるので、'X $\neq 0$ という条件は自動的にみたされることになり、'X / 'X = 1%:P の計算には特に条件が必要ない。よって、'X で約分した後でも 0 での値を計算できる。例えば $D_q(x+a)_q^n=[n](x+a)_q^{n-1}$ という計算には x での約分が必要であるが、多項式として考える場合には上の計算をした後でも 0 での値を求めることができる。この値は本論文の目的である Gauss's binomial formula の証明に必要である。

2. q = 0 のとき高階 D_q が定義できる q = 0 のときに 2 階 D_q を計算してみると

$$\begin{split} (D_q^2 f)(x) &= (D_0^2 f)(x) = (D_0(D_0 f))(x) \\ &= D_0 \left(\lambda x. \, \frac{f(0x) - f(x)}{(0 - 1)x} \right)(x) \\ &= D_0 \left(\lambda x. \, \frac{f(x) - f(0)}{x} \right)(x) \\ &= (D_0 F)(x) \quad (\exists \exists \exists \forall F := \lambda x. \, \frac{f(x) - f(0)}{x} \succeq \exists \forall \forall f \succeq) \\ &= \lambda x. \, \frac{F(x) - F(0)}{x}(x) \\ &= \frac{F(x) - F(0)}{x} \end{split}$$

となるが,

$$F(0) = \frac{f(0) - f(0)}{0} = \frac{0}{0}$$

となってしまう (Coq では 0/0 は 0 と計算されるが, これでも正しい計算結果とはならない). この問題が起きるのはもともとの dq を関数の引数に対して各点ごとに定義しているからで あり、多項式の係数を変化させることで定義している dqp では q=0 でも問題が起きない。 よってこの dqp を用いている Dqp および Dqp'については q=0 かどうかにかかわらず高階の操作を定義できる.

4.4 q-Taylor 展開

この節では、有限次 Taylor 展開の q-類似が成り立つこと、そしてその系として Gauss's binomial formula が成り立つことを示し、形式化する. まず、一般に以下のことが成り立つことを確認しておく.

Theorem 4.4.1 ([1] p5 Theorem 2.1) $\mathbb{K} := \mathbb{R}$ または \mathbb{C} , $V := \mathbb{K}[x]$ とし, D を V 上の線型作用素とする. また, $\{P_n(x)\}_{n=0}$ $\subset V$ $(n=0,1,2,\cdots)$ は次の三条件をみたすとする.

(i)
$$P_0(a) = 1$$
, $P_n(a) = 0$ (${}^{\forall} n \ge 1$)
(ii) $\deg P_n = n$ (${}^{\forall} n \ge 0$)
(iii) $DP_n(x) = P_{n-1}(x)$ (${}^{\forall} n \ge 1$), $D(1) = 0$

ただし, $a \in \mathbb{K}$ である. このとき, 任意の多項式 $f(x) \in V$ に対し, $\deg f(x) = N$ とすると,

$$f(x) = \sum_{n=0}^{N} (D^n f)(a) P_n(x)$$

が成り立つ.

この定理を形式化すると以下のようになる.

```
Theorem general_Taylor D n P (f : {poly R}) a : islinear D \rightarrow isfderiv D P \rightarrow (P 0%N).[a] = 1 \rightarrow (\forall n, (P n.+1).[a] = 0) \rightarrow (\forall m, size (P m) = m.+1) \rightarrow size f = n.+1 \rightarrow f = \sum_(0 \leq i < n.+1) ((D \^ i) f).[a] *: P i.
```

記号の意味などは以下の通りである.

• islinear, isfderiv はそれぞれ

```
Definition islinear (D : {poly R} \rightarrow {poly R}) := \forall a b f g, D ((a *: f) + (b *: g)) = a *: D f + b *: D g. 

Definition isfderiv D (P : nat \rightarrow {poly R}) := \forall n, match n with | 0 \Rightarrow (D (P n)) = 0 | n.+1 \Rightarrow (D (P n.+1)) = P n end.
```

という定義であり, 前者が線形作用素であること, 後者は条件 (iii) を形式化したものである.

• [1] での証明には、 $\{P_0(x), P_1(x), \dots, P_n(x)\}$ が V の基底となることを用いている.これを以下 のように形式化した.

```
Lemma poly_basis n (P : nat \rightarrow {poly R}) (f : {poly R}) : (\forall m, size (P m) = m.+1) \rightarrow (size f \leq n.+1)%N \rightarrow \exists (c : nat \rightarrow R), f = \sum_(0 \leq i < n.+1) c i *: P i.
```

この主張には係数列 c の一意性は含まれていないため, 実際には生成系であることを示している.

この定理において,

$$D \equiv D_q, \quad P_n \equiv \frac{(x-a)_q^n}{[n]!}$$

 $(ただし, n \in \mathbb{Z}_{>0} に対し, [n]! を$

$$[n]! := \begin{cases} 1 & (n = 0) \\ [n] \times [n-1] \times \dots \times [1] & (n \ge 1) \end{cases}$$

と定める)とすることで、有限次 Taylor 展開の q-類似が得られる

Theorem 4.4.2 ([1] p12 Theorem 4.1) f(x) を, N 次の実数係数多項式とする. 任意の $c \in \mathbb{R}$ に対し、

$$f(x) = \sum_{i=0}^{N} (D_q^j f)(c) \frac{(x-c)_q^j}{[j]!}$$

が成り立つ.

Proof. $\frac{(x-a)_q^n}{[n]!}$ が, a, D_q に対して Theorem 4.4.1 の三条件をみたすことを確かめればよい. (i), (ii) は $(x-a)_q^n$ の定義から, (iii) は Proposition 4.2.2 から分かる.

前節で準備した Dqp, qbinom_pos_poly を用いて Theorem 4.4.2 を形式化する.

```
Fixpoint qfact n := match n with \mid 0 \Rightarrow 1 \mid n.+1 \Rightarrow qfact n * qnat n.+1 end.
```

Theorem q_Taylorp n (f : {poly R}) c : $(\forall \ n, \ qfact \ n \neq \emptyset) \rightarrow \\ size \ f = n.+1 \rightarrow \\ f = \sum_{i=1}^{n} (0 \leq i < n.+1) \ ((Dqp' \setminus i) \ f).[c] *: (qbinom_pos_poly c i / (qfact i)%:P).$

1 - (Sun_(0 \(\) 1 \(\) 1 \(\) (Bqb \(\) 1 \(\) 1 \(\) [c] \(\) (qbinon_pos_pory \(\) 1 \(\) (qrace 1)...

Dqp, qbinom_pos_poly をもとの定義に戻したものについては,

```
Theorem q_Taylor n (f : {poly R}) x c : q \neq 0 \rightarrow c \neq 0 \rightarrow (\forall n, q fact n \neq 0) \rightarrow size f = n.+1 \rightarrow f.[x] = \sum_{(Dq ^ i) \# f} c * qbinom_pos c i x / q fact i.
```

Remark 4.4.3 約分のための $c \neq 0$ という条件に加え, 高階 D_q を扱うため前述の通り $q \neq 0$ も必要となる. 具体的には高階 D_{qp} と D_{q} を一致させる補題

```
Lemma hoDqp'_DqE p x n : q \neq 0 \rightarrow x \neq 0 \rightarrow ((Dqp' \^ n) p).[x] = ((Dq \^ n) # p) x. Proof.

move\Rightarrow Hq0 Hx.
rewrite /(_ # _).
elim: n x Hx \Rightarrow [|n IH] x Hx //=.
rewrite Dqp'_DqE // {2}/Dq /dq -!IH //.
by apply mulf_neq0 \Rightarrow //.
Qed.
```

の証明において, IH(Inductive Hypothesis, 帰納の仮定) を使う際に $q*x \neq 0$ という条件が必要となる.

本論文の最後に, x^n と $(x-a)_q^n$ にこの Taylor 展開の q-類似を適用する.

Lemma 4.4.4 ([1] p12 Example (4.4)) $n \in \mathbb{Z}_{>0}$ について,

$$x^{n} = \sum_{j=0}^{n} \begin{bmatrix} n \\ j \end{bmatrix} (x-1)_{q}^{j} \quad \left(\exists \exists \exists \circlearrowleft, \begin{bmatrix} n \\ j \end{bmatrix} := \frac{[n]!}{[j]![n-j]!} \right)$$

が成り立つ.

Proof. Theorem 4.4.2 において, $f(x) = x^n$, c = 1 とする. 任意の正整数 j <= n に対して, $D_q x^n = [n]x^{n-1}$ より,

$$(D_a^j f)(x) = [n][n-1] \cdots [n-j+1]x^{n-j}$$

となるので,

$$(D_q^j f)(1) = [n][n-1] \cdots [n-j+1]$$

が得られる.

Lemma 4.4.5 ([1] p15 Example (5.5)) $n \in \mathbb{Z}_{>0}$ について,

$$(x+a)_q^n = \sum_{j=0}^n \begin{bmatrix} n \\ j \end{bmatrix} q^{j(j-1)/2} a^j x^{n-j}$$

が成り立つ. この式は Gauss's binomial formula と呼ばれる.

Proof. $f = (x + a)_q^n$ とすると,任意の正整数 j <= n に対して,

$$(D_q^j f)(x) = [n][n-1][n-j+1](x+a)_q^{n-j}$$

であり、また

$$(x+a)_q^m = (x+a)(x+qa)\cdots(x+q^{m-1}a)$$

から, $(0+a)_q^m=a\cdot qa\cdots q^{m-1}a=q^{m(m-1)/2}a^m$ となるので,

$$(D_q^j f)(0) = [n][n-1] \cdots [n-j+1] q^{(n-j)(n-j-1)/2} a^{n-j}$$

が成り立つ. よって、Theorem 4.4.2 において、 $f = (x+a)_q^n$ 、c = 0 として、

$$(x+a)_q^n = \sum_{j=0}^n \begin{bmatrix} n \\ j \end{bmatrix} q^{(n-j)(n-j-1)/2} a^{n-j} x^j$$

が得られる. この式の右辺において j をn-j に置き換えることで,

に注意すれば,

$$(x-a)_q^n = \sum_{j=0}^n \begin{bmatrix} n \\ j \end{bmatrix} q^{j(j-1)/2} a^j x^{n-j}$$

が成り立つ.

この二つの等式の形式化はそれぞれ次の通り.

```
Lemma q_Taylorp_pow n: (\forall n, qfact n \neq 0) \rightarrow 'X^n = \sum_(0 \leq i < n.+1) (qbicoef n i *: qbinom_pos_poly 1 i).

Definition qbicoef n j := qfact n / (qfact j * qfact (n - j)).

Theorem Gauss_binomial' a n : (\forall n, qfact n \neq 0) \rightarrow qbinom_pos_poly (-a) n = \sum_(0 \leq i < n.+1) (qbicoef n i * q ^+ ((n - i) * (n - i - 1))./2 * a ^+ (n - i)) *: 'X^i.

Theorem Gauss_binomial a n : (\forall n, qfact n \neq 0) \rightarrow qbinom_pos_poly (-a) n = \sum_(0 \leq i < n.+1) (qbicoef n i * q ^+ (i * (i - 1))./2 * a ^+ i) *: 'X^(n - i).

Gauss_binomial' はjをn - jに置き換える前の等式である.
```

Remark 4.4.6 Gauss_binomial'は q_Taylorp において c = 0 として証明している. q_Taylorp では約分の計算をしているが、多項式を用いて定義しているため 0 での値を計算できる.

参考文献

- [1] Victor Kac, Pokman Cheung, Quantum Calculus, Springer, 2001.
- [2] https://github.com/nakamurakaoru/q-analogue
- [3] 萩原 学/アフェルト・レナルド, Cog/SSReflect/Mathcomp, 森北出版, 2018
- [4] https://coq.inria.fr/distrib/current/stdlib/
- [5] https://github.com/math-comp/math-comp
- [6] https://github.com/math-comp/analysis
- [7] 梅村浩,『楕円関数論楕円曲線の解析学』,東京大学出版会,2000.
- [8] H.P.Barendregt, Lambda Calculi with Types